

A Machine-Checked Correctness Proof of  
Normalization by Evaluation for Simply Typed  
Lambda Calculus

Author: András Kovács      Advisor: Ambrus Kaposi

Budapest, 2017

# Contents

<b>1</b>	<b>Introduction</b>	<b>4</b>
1.1	Overview . . . . .	4
1.2	Related Work . . . . .	5
<b>2</b>	<b>Metatheory</b>	<b>7</b>
2.1	Type Theory . . . . .	7
2.2	Agda . . . . .	8
2.2.1	Basic Constructions . . . . .	8
2.2.2	Standard Library . . . . .	11
<b>3</b>	<b>Syntax</b>	<b>14</b>
3.1	Base Syntax . . . . .	14
3.2	Context Embeddings and Substitutions . . . . .	16
3.3	Conversion . . . . .	20
<b>4</b>	<b>Normalization</b>	<b>21</b>
4.1	Preliminaries: Standard Model . . . . .	21
4.2	Normalization with a Kripke Model . . . . .	21
4.3	Laws for Embedding and Substitution . . . . .	21
4.4	Completeness . . . . .	22
4.5	Naturality Proofs . . . . .	22
4.6	Soundness . . . . .	22
4.7	Stability . . . . .	22

<i>CONTENTS</i>	3
<b>5 Variations</b>	<b>23</b>
5.1 Direct Presheaf Model . . . . .	23
5.2 More Efficient Evaluation . . . . .	23
<b>6 Discussion and Future Work</b>	<b>24</b>

# Chapter 1

## Introduction

### 1.1 Overview

Normalization animates the syntax of typed  $\lambda$ -calculi, making them useful as programming languages. Also, in contrast to general terms, normal forms possess a more restricted structure which simplifies formal reasoning. Moreover, normalization allows one to decide convertibility of terms, which is required during type checking polymorphic and dependent type theories.<sup>1</sup>

This thesis presents an efficient and verified implementation of  $\beta\eta$ -normalization for simply typed  $\lambda$ -calculus (STLC) in Agda, an implementation of constructive type theory. In this setting, the syntax can be seen as an embedded language, and manipulating embedded terms is a limited form of metaprogramming. Normalization can be reused in the implementation of proof tactics, decision procedures or domain-specific languages, therefore efficiency is desirable. It is also preferable that the syntax remains as simple and self-contained as possible, both for pedagogical purposes and for the ease of reuse in other Agda developments. For this reason we choose an intrinsic syntax with de Bruijn indices and implicit substitutions.

---

<sup>1</sup>In particular, the types of System  $F_\omega$  largely correspond to simply typed  $\lambda$ -terms.

We are not concerned with mapping the algorithm to lower-level abstract machines or hardware. Thus we are free to make use of the most convenient evaluation mechanism at hand: that of the metalanguage. This way, we can gloss over implementation details of efficient higher-order evaluation. This is the core idea of normalization by evaluation (NbE) from an operational point of view. Also, totality of evaluation in the metatheory is always implicitly assumed from the inside. This lets us implement normalization in a structurally recursive way, making its totality trivial. NbE also naturally supports  $\eta$ -normalization.<sup>2</sup>

If we are to trust a particular normalization algorithm, we need to prove the following properties (CIT: bigstep or alti):

- *Completeness*: terms are convertible to their normal forms.
- *Soundness*: normalization takes convertible terms to the same normal form.

Additionally, we may require stability, which establishes that there is no redundancy in normal forms:

- *Stability*: normalization acts as the identity function on normal terms.

TODO: Summary of chapters, some blahblah (at least half page worth of it)

## 1.2 Related Work

TODO: Not directly related background: Martin-Löf, Berger & Schwichtenberg. (one paragraph)

Abel’s CIT work gives a comprehensive overview of NbE, and also develops it for a range of type theories. The approach differs markedly from ours, as it uses on extrinsic syntax and separate typed and untyped phases of evaluation.

Catarina Coquand’s work CIT is the most closely related to our development. She

---

<sup>2</sup>In contrast to approaches based on small-step reduction, where  $\eta$  requires complicated setup.

formally proves soundness and completeness of NbE for simple type theory, however she considers a nameful syntax with explicit substitutions while we use intrinsic de Bruijn variables with implicit substitution.

Altenkirch and Chapman CIT formalize a big-step normalization algorithm in Agda. This development uses an explicit substitution calculus and implements normalization using an environment machine with first-order closures. The algorithm works similarly to ours on the common syntactic fragment, but it is not structurally recursive and hence specifies evaluation as a inductive relation, using the Bove-Capretta method CIT.

Altenkirch and Kaposi CIT use a glued presheaf model (“presheaf logical predicate”) for NbE for a minimal dependent type theory. This development provided the initial inspiration for the formalization in this thesis; it seemed plausible that “scaling down” dependently typed NbE to simple type theory would yield a formalization that is more compact than existing ones. This turned to be the case; however the discussion of resulting development is relegated to sec. 5.1, because using a Kripke model has the advantage of clean separation of the actual algorithm and its naturality proofs, which was deemed preferable to the somewhat less transparent presheaf construction. Also, our work uses a simple presheaf model rather than the glued model used in Ibid. or in the work of Altenkirch, Hoffman and Streicher CIT.

# Chapter 2

## Metatheory

In this chapter the metatheory used for formalization is presented, first broadly, then its specific implementation in Agda.

### 2.1 Type Theory

The basic system we use is intensional Martin-Löf type theory (MLTT). For an overview and tutorial see the first chapter of the Homotopy Type Theory book [14]. However, there is no canonical definition of intensional type theory. There are numerous variations concerning type universes and the range of allowed type definitions. Our development uses the following features:

- *Strictly positive inductive definitions* CIT. These are required for an intrinsically well-typed syntax. We do not require induction-recursion, induction-induction, higher induction or large inductive types.
- *Two predicative universes, named  $\mathbf{Set}_0$  and  $\mathbf{Set}_1$ , with large elimination into  $\mathbf{Set}_0$ .*  $\mathbf{Set}_0$  is the base universe in Agda, i. e. the universe of small types. Large elimination is needed for recursive definitions of semantic types, since inductive

definitions for them would not be positive and hence would be illegal. We only need `Set1` to type the large eliminations. In Agda `Set` is a synonym for `Set0`.

- *Function extensionality as an axiom.* We could eschew it by using setoid reasoning with semantic equivalence, as C. Coquand does CIT. However, we only use function extensionality in correctness proofs, and the normalization function does not refer to it or to any other postulate. We are content with *logical* as opposed to computational content for correctness proofs. This way, the numerous congruence lemmas needed for setoid reasoning can be skipped. Also, function extensionality has computational interpretation in cubical CIT and observational CIT type theories, to which this development could be plausibly ported in the future (when practical implementations of the mentioned theories become available).

In Agda 2.5.2, there is robust support for dependent pattern matching without Streicher’s K axiom CIT. We use the `--without-K` language option for all of our code in the main development. However, we do use axiom K in sec. 5.1 for the direct presheaf model, but only for technical convenience, as it can be avoided by additional uniqueness proofs for equalities.

## 2.2 Agda

Agda is a dependently typed programming language and proof assistant. Its design and core features are described in Ulf Norell’s thesis [10]. The latest documentation is available online [6]. For a book-length introduction geared towards beginners, see [13]. We nonetheless summarize here the core constructions.

### 2.2.1 Basic Constructions

NOTE: EITHER CULL THIS PART OR EXTEND IT. Ideally it should be culled, because a few page worth of Agda intr isn’t realistically useful.



Two essential artifacts in Agda code are *inductive type definitions* and *function definitions*.

*Inductive definitions* introduce new types. Agda has *open type universes*, which means that programmers may freely add new types as long as they preserve logical consistency<sup>1</sup>. An inductive definition involves a type constructor declaration followed by zero or more term constructor declarations. For example, the type of natural numbers is defined the following way:

```
data ℕ : Set where
  zero : ℕ
  suc   : ℕ → ℕ
```

Inductive definitions may have *parameters* and *indices*. The former are implicitly quantified over the term constructors, but must be uniform in the constructors' return types. The latter must be explicitly quantified in term constructors, but are allowed to vary. The definition for length-indexed vectors exhibits both:

```
data Vec (A : Set) : ℕ → Set where
  nil   : Vec A zero
  cons  : {n : ℕ} → A → Vec A n → Vec A (suc n)
```

In a type constructor declaration, parameters are listed left to the colon, while indices are to the right. `A : Set` is a parameter, so it is implicitly quantified and is the same in the return types of `nil` and `cons`. In contrast, the length index is quantified in `cons`. We can use brackets to make parameters implicit; here `cons` has an implicit first parameter, and can be used like `cons zero nil` for a value of `Vec ℕ (suc zero)`. Implicit arguments are filled in by Agda's unification algorithm. Alternatively, the `∀` symbol can be used to leave the types of parameters implicit, and we could define `cons` as follows:

```
cons : ∀ {n} → A → Vec A n → Vec A (suc n)
```

*Function definitions* are given by pattern matching. In Agda, pattern matching implements branching evaluation as it is usual in functional languages, but it also refines type indices based on particular selected cases. The main practical difference

---

<sup>1</sup>Agda checks *strict positivity* as a sufficient condition for consistency.

between parameters and indices is that we can gain information about indices by pattern matching, while we can't infer anything about parameters just by having a parameterized term.

For example, in the following definition of concatenation for **Vec**, the result type is refined depending on whether the first argument is empty:

```

_+_ : ℕ → ℕ → ℕ
zero + m = m
suc n + m = suc (n + m)

_++_ : ∀ {A n m} → Vec A n → Vec A m → Vec A (n + m)
nil      ++ ys = ys
cons x xs ++ ys = cons x (xs ++ ys)

```

In the `cons x xs ++ ys` case, the result type is refined to `Vec A (suc (n + m))`, allowing us to build a `suc`-long result on the right hand side with `cons`.

Agda function definitions must be total, hence pattern matching must be exhaustive and recursive calls must be structurally decreasing. Dependent pattern matching with structural recursion allows us to write the same proofs as inductive eliminators, but with a more convenient interface CIT. Also, injectivity and disjointness of term constructors is implicit in pattern matching, while they have to be separately proven when using eliminators.

As a notational convention, we shall take cues from the Idris [4] programming language and sometimes leave implicit parameters implicit even in type declarations of functions or constructors. In this style, declaring `_++_` looks as follows:

```

_++_ : Vec A n → Vec A m → Vec A (n + m)

```

This is not valid Agda, but we shall do this whenever the types and binding status of parameters are obvious.

### 2.2.2 Standard Library

Our development does not have any external dependencies. We need less than a hundred lines of code from the Agda standard library CIT, but we choose to include it along the project in `Lib.agda`, for portability, and also make some changes.

Most of `Lib.agda` concerns equality and equational reasoning. Propositional equality is the same as in standard Agda:

```
data _≡_ {i}{A : Set i} (x : A) : A → Set i where
  refl : x ≡ x
```

We also postulate function extensionality, for functions with both implicit and explicit arguments (the “i” in `fexti` stands for “implicit”):

```
postulate
  fext  : (∀ x → f x ≡ g x) → f ≡ g
  fexti : (∀ x → f {x} ≡ g {x}) → (λ {x} → f {x}) ≡ (λ {x} → g {x})
```

Note that the types of `f` and `g` are left implicit above, in accordance with our notational liberties. In full Agda we write:

```
fext :
  ∀ {i j}{A : Set i}{B : A → Set j}{f g : (x : A) → B x}
  → ((x : A) → f x ≡ g x) → f ≡ g
```

Here the quantification noise is increased by universe indices. Universe polymorphism is only used in `Lib.agda`, but we will henceforth omit universe indices anyway.

Our style of equational reasoning deviates from the standard library. First, we use a more compact notation for symmetry and transitivity, inspired by the Homotopy Type Theory book [14]:

```
_■_ : x ≡ y → y ≡ z → x ≡ z -- transitivity
refl ■ refl = refl

_⁻¹ : x ≡ y → y ≡ x -- symmetry
refl ⁻¹ = refl
```

$\_^{-1}$  is postfix, so if we have  $(p : x \equiv y)$ , then  $(p^{-1})$  has type  $(y \equiv x)$ .

Second, for reasoning about congruences we mainly use two operations. The first (named `ap` in the HoTT book and `cong` in the standard library) expresses that functions respect equality:

```
_&_ : (f : A → B) → x ≡ y → f x ≡ f y
f & refl = refl
infixl 9 _&_
```

The second one expresses that *non-dependent function application* respects equality as well:

```
_⊗_ : f ≡ g → x ≡ y → f x ≡ g y
refl ⊗ refl = refl
infixl 8 _⊗_
```

Using these two operators, we can lift non-dependent functions with arbitrary arities to congruences. For example, if we have  $(p : a_0 \equiv a_1)$ ,  $(q : b_0 \equiv b_1)$  and  $(f : A \rightarrow B \rightarrow C)$ , then  $(f \& p \otimes q)$  has type  $(f a_0 b_0 \equiv f a_1 b_1)$ . Note that both operators associate left, and `_&_` binds more strongly. This is reminiscent of the lifting syntax with applicative functors [9] in Haskell programming.

Furthermore, we use coercion instead of transport (see HoTT book [14, pp. 72], also named `subst` in the Agda standard library CIT):

```
coe : A ≡ B → A → B
coe refl a = a
```

`transport/subst` can be recovered by `_&_` and `coe`:

```
subst : (P : A → Set) → x ≡ y → P x → P y
subst P eq = coe (P & eq)
```

The choice for `coe` is mainly stylistic and rather subjective. `coe (P & eq)` does not look worse than `subst P eq`, but `coe` by itself is used often and is more compact than `subst id`.

In Agda, sometimes an explicit “equational reasoning” syntax is used. For example,

commutativity for addition may look like the following:

```

+-comm : (m n : ℕ) → m + n ≡ n + m
+-comm zero    n = +-right-identity n -1
+-comm (suc m) n = begin
  suc m + n    ≡{ }
  suc (m + n)  ≡{ suc & +-comm m n }
  suc (n + m)  ≡{ +-suc n m -1 }
  n + suc m    QED

```

In formal development we do not use this style and instead keep the intermediate expressions implicit:

```

+-comm (suc m) n = (suc & +-comm n m) ■ (+-suc n m -1)

```

This is because in an interactive environment one can step through the intermediate points fairly easily, and writing them out adds significant visual clutter. However, we will use informal equational reasoning when it aids explanation.

We also borrow  $\Sigma$  (dependent sum),  $\tau$  (unit type),  $\perp$  (empty type) and  $\_ \sqcup \_$  (disjoint union) from the standard library. Elements of  $\Sigma$  are constructed as  $(a, b)$  pairs, and the projections are `proj1` and `proj2`.  $\tau$  has `tt` (“trivially true”) as sole element. Injection into disjoint sums is done with `inj1` and `inj2`. The *ex falso* principle is named `⊥-elim`, and we have `⊥-elim : {A : Set} → ⊥ → A`.

# Chapter 3

## Syntax

In this chapter we present the syntax of simply typed lambda calculus - in our case intrinsic syntax with implicit substitutions and de Bruijn indices. We take term conversion to be part of the syntax and therefore include it in this chapter.<sup>1</sup> Since the definition of conversion refers to substitution and weakening, they are also presented here.

### 3.1 Base Syntax

The complete definition is the following:

```
data Ty : Set where
  ι      : Ty
   $\_ \Rightarrow \_$  : Ty → Ty → Ty

data Con : Set where
  •      : Con
   $\_ , \_$  : Con → Ty → Con
```

---

<sup>1</sup>We take the view that the “proper” definition of syntax would be an initial object in some suitable category of models, e. g. the initial category with families. Thus, models include conversion as propositional equations, so the syntax is quotiented by conversion as well; see e. g. [2]. We do not use this definition because quotients are not yet natively supported in Agda, and we find value in developing a conventional presentation first.

```

data _∈_ (A : Ty) : Con → Set where
  vz : A ∈ (Γ , A)
  vs : A ∈ Γ → A ∈ (Γ , B)

data Tm Γ : Ty → Set where
  var : A ∈ Γ → Tm Γ A
  lam : Tm (Γ , A) B → Tm Γ (A ⇒ B)
  app : Tm Γ (A ⇒ B) → Tm Γ A → Tm Γ B

```

There is a base type  $\iota$  (Greek iota) and function types. Contexts are just lists of types.  $\_ \in \_$  is an inductively defined membership relation on contexts. It has the same structure as de Bruijn indices; for this reason they are named **vz** for “zero” and **vs** for “suc”. Terms are parameterized by a  $\Gamma$  context and indexed by a type. **var** picks an entry from the contexts, **lam** is abstraction while **app** is function application.

As an example, the identity function on the base type is represented as:

```

id-ι : Tm • (ι ⇒ ι)
id-ι = lam (var vz) -- with nameful syntax: λ (x : ι) . x

```

The presented syntax is *intrinsic*. In other words, only the well-typed terms are defined. An *extrinsic* definition would first define untyped *preterms* and separately a typing relation on preterms and contexts:

```

data Tm : Set where -- omitted constructors
data _⊢_ (Γ : Con) → Tm → Ty → Set where -- omitted constructors

```

Now,  $\Gamma \vdash t \in A$  expresses that the preterm  $t$  is well-formed and has type  $A$  in context  $\Gamma$ .

There are advantages and disadvantages to both intrinsic and extrinsic definitions. With intrinsic syntax, well-formedness and type preservation come for free. On the other hand, proofs and computations which do not depend on types are easier to formulate with preterms. In the case of STLC, the type system is simple enough so that carrying around type information never becomes burdensome, so in this setting intrinsic definitions are all-around more convenient. For polymorphic systems such as System

$F$ , intrinsic typing introduces significant bureaucratic overhead, by tagging terms with type coercions; see Benton et al. [3] for approaches to dealing with them. For extrinsic syntax, powerful automation for reasoning about substitution is available in Coq, allowing spectacularly compact proofs [12]. It remains to be seen if it can be practically adapted to intrinsic syntax. For dependent type theories, intrinsic typing with quotient inductive-inductive definitions seems to be most promising, as it requires a relatively compact set of rules [2], in contrast to extrinsic approaches which suffer from a veritable explosion of well-formedness conditions and congruence rules.

## 3.2 Context Embeddings and Substitutions

Another salient feature of our syntax is the lack of explicit substitution, i. e. we define substitutions separately as lists of terms, and their action on terms is given as a recursive function. We choose implicit substitution for two reasons:

1. It is the canonical presentation of STLC in textbooks [11, 7].
2. Lighter formalization. With implicit substitution, the definition of conversion becomes significantly smaller. With explicit substitutions, we need congruence closure of  $\beta$  and  $\eta$  conversion on terms and substitutions, plus rules for the action of substitution on terms. With implicit substitutions, only terms need to be considered. For illustration, Altenkirch and Chapman’s definition[1] of conversion for explicit substitution calculus has twenty-four rules, while this development has only eight. It also follows that with our syntax, equational reasoning about substitutions involves definitional or propositional equalities, which are automatically respected by all constructions. There is minor complication involving semantic interpretation of substitutions, discussed in sec. 4.6.

At this point, we are concerned with conversion for terms, and substitution is introduced in order to define  $\beta$ -conversion. For this we need single substitution. However, simultaneous substitution is easier to define and reason about. Thus, we shall define



the former using the latter and identity substitutions.

Identity substitutions are those which have the identity action on terms. Conceptually, they assign to each free variable in a term the same variable. They can be implemented in a number of ways. One possibility is to make identity substitutions explicit constructions, in which case their action on terms can be simply defined as identity. However, we instead opt for a *normalized* representation of substitutions, as lists of terms:

```
data Sub (Γ : Con) : Con → Set where
  •   : Sub Γ •
  _,_ : Sub Γ Δ → Tm Γ A → Sub Γ (Δ , A)
```

Elements of  $\text{Sub } \Gamma \Delta$  contain a  $\text{Tm } \Gamma A$  for each  $A$  type in  $\Delta$ . In other words, an element of  $\text{Sub } \Gamma \Delta$  assigns to each variable in  $\Delta$  a term in  $\Gamma$ . Following McBride [8, pp. 24] we could give an alternative  $\text{Sub}$  definition, which makes the underlying assignment obvious:

```
Sub : Con → Con → Set
Sub Γ Δ = {A : Ty} → A ∈ Δ → Tm Γ A
```

However, this representation contains many definitionally distinct functions with the same action on terms. Thus, it is larger than what is strictly necessary, which may or may not result in technical difficulties. We aim to choose the smallest representation that does the job, as a general principle.

Before we can define identity substitutions and the action of substitution on terms, we need to express the notion that whenever one has a term in a context, one can construct essentially the same term in a larger context. Specifically here, we need an operation with type  $(\forall \{A \in \Gamma \Delta\} \rightarrow \text{Sub } \Gamma \Delta \rightarrow \text{Sub } (\Gamma , A) (\Delta , A))$ , in order to be able to push substitutions under  $\lambda$ -binders when recursing on terms. Constructing this substitution requires that we embed all terms in the input substitution into the extended  $(\Gamma , A)$  context, then extend the result with `var vz`.

We define *order-preserving embeddings* the following way:

```
data OPE : Con → Con → Set where
```

$\bullet : \text{OPE } \Gamma \Delta \rightarrow \text{OPE } (\Gamma, A) \Delta$   
 $\text{drop} : \text{OPE } \Gamma \Delta \rightarrow \text{OPE } (\Gamma, A) \Delta$   
 $\text{keep} : \text{OPE } \Gamma \Delta \rightarrow \text{OPE } (\Gamma, A) (\Delta, A)$

Elements of  $\text{OPE } \Gamma \Delta$  express that  $\Delta$  can be obtained from  $\Gamma$  by dropping zero or more entries in order.

Some choices can be made here. C. Coquand [5] uses an explicit identity embedding constructor with type  $(\forall \{\Gamma\} \rightarrow \text{OPE } \Gamma \Gamma)$  instead of our  $\bullet$ . This slightly simplifies some proofs and slightly complicates others. It does not have propositionally unique identity embeddings<sup>2</sup>, but overall the choice between this definition and ours is fairly arbitrary.

An alternative solution is using *renamings*. Renamings are essentially substitutions containing only variables. They can fulfill the same role as order-preserving embeddings. However, they are larger than necessary, since they can represent permutations of context entries as well, which we have no need for.

Embeddings have action on variables and terms, reconstructing them in larger contexts. We denote embedding by lowercase “e” subscripts:

$\epsilon_e : \text{OPE } \Gamma \Delta \rightarrow A \in \Delta \rightarrow A \in \Gamma$   
 $\epsilon_e \bullet \quad v \quad = v$   
 $\epsilon_e (\text{drop } \sigma) v \quad = v \sigma (\epsilon_e \sigma v)$   
 $\epsilon_e (\text{keep } \sigma) v z \quad = v z$   
 $\epsilon_e (\text{keep } \sigma) (v \sigma v) = v \sigma (\epsilon_e \sigma v)$   
  
 $\text{Tm}_e : \text{OPE } \Gamma \Delta \rightarrow \text{Tm } \Delta A \rightarrow \text{Tm } \Gamma A$   
 $\text{Tm}_e \sigma (\text{var } v) \quad = \text{var } (\epsilon_e \sigma v)$   
 $\text{Tm}_e \sigma (\text{lam } t) \quad = \text{lam } (\text{Tm}_e (\text{keep } \sigma) t)$   
 $\text{Tm}_e \sigma (\text{app } f a) = \text{app } (\text{Tm}_e \sigma f) (\text{Tm}_e \sigma a)$

Identity embeddings keep every entry:

$\text{id}_e : \forall \{\Gamma\} \rightarrow \text{OPE } \Gamma \Gamma$   
 $\text{id}_e \{\bullet\} \quad = \bullet$   
 $\text{id}_e \{\Gamma, A\} = \text{keep } (\text{id}_e \{\Gamma\})$

We define *wk* as shorthand for the embedding that only drops the topmost entry:

---

<sup>2</sup>Since wrapping the identity embedding with any number of *keep*-s also yields identity embeddings.

$wk : \forall \{A \ \Gamma\} \rightarrow OPE \ (\Gamma, A) \ \Gamma$   
 $wk = drop \ id_e$

Embedding of terms can be pointwise extended to substitutions:

$\_s \circ_e \_ : Sub \ \Delta \ \Sigma \rightarrow OPE \ \Gamma \ \Delta \rightarrow Sub \ \Gamma \ \Sigma$   
 $\bullet \quad \_s \circ_e \ \delta = \bullet$   
 $(\sigma, t) \_s \circ_e \ \delta = (\sigma \_s \circ_e \ \delta), Tm_e \ \delta \ t$

The notation  $\_s \circ_e \_$  expresses that it is a composition of a substitution and an embedding. Its type resembles that of ordinary function composition, if we consider **Sub** and **OPE** as context morphisms. We will expand on the categorical interpretation of this in sec. 4.3.

There is a canonical injection  $\ulcorner \_ \urcorner_{ope}$  from **OPE** to **Sub**:

$drop_s : Sub \ \Gamma \ \Delta \rightarrow Sub \ (\Gamma, A) \ \Delta$   
 $drop_s \ \sigma = \sigma \_s \circ_e \ wk$   
  
 $keep_s : Sub \ \Gamma \ \Delta \rightarrow Sub \ (\Gamma, A) \ (\Delta, A)$   
 $keep_s \ \sigma = drop_s \ \sigma, \text{var } vz$   
  
 $\ulcorner \_ \urcorner_{ope} : OPE \ \Gamma \ \Delta \rightarrow Sub \ \Gamma \ \Delta$   
 $\ulcorner \bullet \urcorner_{ope} = \bullet$   
 $\ulcorner drop \ \sigma \urcorner_{ope} = drop_s \ \ulcorner \sigma \urcorner_{ope}$   
 $\ulcorner keep \ \sigma \urcorner_{ope} = keep_s \ \ulcorner \sigma \urcorner_{ope}$

$keep_s$  is precisely the substitution operation we were looking for. Now the action of substitution on terms and variables can be defined:

$\epsilon_s : Sub \ \Gamma \ \Delta \rightarrow A \in \Delta \rightarrow Tm \ \Gamma \ A$   
 $\epsilon_s \ (\sigma, t) \ vz = t$   
 $\epsilon_s \ (\sigma, t) \ (vs \ v) = \epsilon_s \ \sigma \ v$   
  
 $Tm_s : Sub \ \Gamma \ \Delta \rightarrow Tm \ \Delta \ A \rightarrow Tm \ \Gamma \ A$   
 $Tm_s \ \sigma \ (\text{var } v) = \epsilon_s \ \sigma \ v$   
 $Tm_s \ \sigma \ (\text{lam } t) = \text{lam } (Tm_s \ (keep_s \ \sigma) \ t)$   
 $Tm_s \ \sigma \ (\text{app } f \ a) = \text{app } (Tm_s \ \sigma \ f) \ (Tm_s \ \sigma \ a)$

$\epsilon_s$  simply looks up a term from a substitution<sup>3</sup>, while  $Tm_s$  recurses into terms to

---

<sup>3</sup>The immediate reason for not naming  $\epsilon_s$  simply “lookup” is that we will need several different lookup functions for various purposes. For any sizable formal development, naming schemes eventually

substitute all variables. Identity substitutions can be defined as well:

```

ids : {Γ : Con} → Sub Γ Γ
ids {•}      = •
ids {Γ , A} = keeps ids

```

Single substitution with a  $(t : \text{Tm } \Gamma \text{ A})$  term is given by  $(\text{id}_s, t)$ . This assigns the  $t$  term to the zeroth de Bruijn variable and leaves all other variables unchanged.

### 3.3 Conversion

The conversion relation is given as:

```

data _~_ {Γ} : ∀ {A} → Tm Γ A → Tm Γ A → Set where
  η      : t ~ lam (app (Tme wk t) (var vz))
  β      : app (lam t) t' ~ Tms (ids , t') t

  lam    : t ~ t' → lam t ~ lam t'
  app1  : f ~ f' → app f a ~ app f' a
  app2  : a ~ a' → app f a ~ app f a'

  ~refl  : t ~ t
  _~-1  : t ~ t' → t' ~ t
  _~■_   : t ~ t' → t' ~ t'' → t ~ t''

```

$\eta$  and  $\beta$  are the actual conversion rules. We push  $t$  under a lambda with  $(\text{Tm}_e \text{ wk})$  in  $\eta$ , and use single substitution for  $\beta$ . The rest are rules for congruence ( $\text{lam}$ ,  $\text{app}_1$ ,  $\text{app}_2$ ) and equivalence closure ( $\sim\text{refl}$ ,  $\sim^{-1}$ ,  $\sim^\bullet$ ).

---

emerge, for better or worse. The author has found that uniformly encoding salient information about types of operations in their names is worthwhile to do. It is also not easy to hit the right level of abstraction; too little and we repeat ourselves too much or neglect to include known structures, but too much abstraction may cause the project to be less accessible and often also less convenient to develop in the first place. Many of the operations and proofs here could be defined explicitly using categorical language, i. e. bundling together functors' actions on objects, morphisms and category laws in records, then projecting out required components. Our choice is to keep the general level of abstraction low for this development.

# Chapter 4

## Normalization

TODO

### 4.1 Preliminaries: Standard Model

TODO

### 4.2 Normalization with a Kripke Model

TODO

### 4.3 Laws for Embedding and Substitution

TODO

## 4.4 Completeness

TODO

## 4.5 Naturality Proofs

TODO

## 4.6 Soundness

TODO

## 4.7 Stability

TODO

# Chapter 5

## Variations

### 5.1 Direct Presheaf Model

- Definition, proofs
- Semantic equality == propositional equality
- However: funext in definition of nf. Use external canonicity proof or metatheory with funext

### 5.2 More Efficient Evaluation

- efficient embedding for semantic contexts
- Agda benchmarking
- performance limitations of intrinsic syntax compared to type assignment

# Chapter 6

## Discussion and Future Work

- Technical overhead: Vs big-step, hereditary subst, Coquand, Abel (?)
- Linecounts vs big-step and hsubst (email C. Coquand to get sources?)
- Usability as EDSL normalizer
- Scaling up the technique
- With implicit substitution and conversion relation: to System F, probably



# Bibliography

- [1] Thorsten Altenkirch and James Chapman. “Big-step normalisation”. In: *Journal of Functional Programming* 19.3-4 (2009), pp. 311–333.
- [2] Thorsten Altenkirch and Ambrus Kaposi. “Type theory in type theory using quotient inductive types”. In: *ACM SIGPLAN Notices*. Vol. 51. 1. ACM. 2016, pp. 18–29.
- [3] Nick Benton et al. “Strongly typed term representations in Coq”. In: *Journal of automated reasoning* 49.2 (2012), pp. 141–159.
- [4] Edwin Brady. “Idris, a general-purpose dependently typed programming language: Design and implementation”. In: *Journal of Functional Programming* 23.05 (2013), pp. 552–593.
- [5] Catarina Coquand. “A proof of normalization for simply typed lambda calculus written in ALF”. In: *Proceedings of the Workshop on Types for Proofs and Programs*. Citeseer. 1992, pp. 85–92.
- [6] Agda developers. *Agda documentation*. 2017. URL: <https://agda.readthedocs.io/en/v2.5.2/> (visited on 03/10/2017).
- [7] Robert Harper. *Practical foundations for programming languages*. Cambridge University Press, 2016.
- [8] Conor McBride. “Datatypes of Datatypes”. In: (2015).
- [9] Conor McBride and Ross Paterson. “Applicative programming with effects”. In: *Journal of functional programming* 18.01 (2008), pp. 1–13.

- [10] Ulf Norell. *Towards a practical programming language based on dependent type theory*. Vol. 32. Citeseer, 2007.
- [11] Benjamin C Pierce. *Types and programming languages*. MIT press, 2002.
- [12] Steven Schäfer, Tobias Tebbi, and Gert Smolka. “Autosubst: Reasoning with de Bruijn Terms and Parallel Substitutions”. In: *Interactive Theorem Proving - 6th International Conference, ITP 2015, Nanjing, China, August 24-27, 2015*. Ed. by Xingyuan Zhang and Christian Urban. LNAI. Springer-Verlag, Aug. 2015.
- [13] Aaron Stump. *Verified Functional Programming in Agda*. New York, NY, USA: Association for Computing Machinery and Morgan & Claypool, 2016. ISBN: 978-1-97000-127-3.
- [14] The Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathematics*. Institute for Advanced Study: <https://homotopytypetheory.org/book>, 2013.