

A Machine-Checked Correctness Proof of
Normalization by Evaluation for Simply Typed
Lambda Calculus

Author: András Kovács Advisor: Ambrus Kaposi

Budapest, 2017

Contents

1	Introduction	4
1.1	Overview	4
1.2	Related Work	5
2	Metatheory	7
2.1	Type Theory	7
2.2	Agda	8
2.2.1	Basic Constructions	9
2.2.2	Standard Library	11
3	Syntax	14
3.1	Base Syntax	14
3.2	Context Embeddings and Substitutions	16
3.2.1	Embeddings	16
3.2.2	Substitutions	18
3.3	Conversion	20
4	Normalization	22
4.1	Normal terms	22
4.2	Preliminaries: Standard and Enriched Models	23
4.3	Normalization	25
4.4	Laws for Embedding and Substitution	29
4.5	Completeness	29

<i>CONTENTS</i>	3
4.6 Presheaf Model	29
4.7 Soundness	29
4.8 Stability	30
5 Variations	31
5.1 Direct Presheaf Model	31
5.2 More Efficient Evaluation	31
6 Discussion and Future Work	32

Chapter 1

Introduction

1.1 Overview

Normalization animates the syntax of typed λ -calculi, making them useful as programming languages. Also, in contrast to general terms, normal forms possess a more restricted structure which simplifies formal reasoning. Moreover, normalization allows one to decide convertibility of terms, which is required during type checking polymorphic and dependent type theories.¹

This thesis presents an efficient and verified implementation of $\beta\eta$ -normalization for simply typed λ -calculus (STLC) in Agda, an implementation of constructive type theory. In this setting, the syntax can be seen as an embedded language, and manipulating embedded terms is a limited form of metaprogramming. Normalization can be reused in the implementation of proof tactics, decision procedures or domain-specific languages, therefore efficiency is desirable. It is also preferable that the syntax remains as simple and self-contained as possible, both for pedagogical purposes and for the ease of reuse in other Agda developments. For this reason we choose an intrinsic syntax with de Bruijn indices and implicit substitutions.

¹In particular, the types of System F_ω largely correspond to simply typed λ -terms.

We are not concerned with mapping the algorithm to lower-level abstract machines or hardware. Thus we are free to make use of the most convenient evaluation mechanism at hand: that of the metalanguage. This way, we can gloss over implementation details of efficient higher-order evaluation. This is the core idea of normalization by evaluation (NbE) from an operational point of view. Also, totality of evaluation in the metatheory is always implicitly assumed from the inside. This lets us implement normalization in a structurally recursive way, making its totality trivial. NbE also naturally supports η -normalization.²

If we are to trust a particular normalization algorithm, we need to prove the following properties (we borrow terminology from [2]):

- *Completeness*: terms are convertible to their normal forms.
- *Soundness*: normalization takes convertible terms to the same normal form.

Additionally, we may require stability, which establishes that there is no redundancy in normal forms:

- *Stability*: normalization acts as the identity function on normal terms.

Chapter 2 describes the metatheory we work in. In Chapter 3, we formalize the syntax of STLC along with $\beta\eta$ -conversion. Chapter 4.3 first presents the normalization function, then proceeds with its correctness proofs. In Chapter 5 we describe two alternative formalizations, one with more efficient evaluation and one with more concise correctness proofs.

1.2 Related Work

Martin-Löf [20] first used NbE (though the term has not been coined yet) to show decidability of type checking for the 1975 version of his intuitionistic type theory. Berger and Schwichtenberg reintroduced NbE in 1991 [9] as a normalization method for lambda

²In contrast to approaches based on small-step reduction, where η requires complicated setup.

calculi. Abel’s work [1] gives a comprehensive overview of NbE, and also develops it for a range of type theories. The approach differs markedly from ours, as it uses extrinsic syntax and separate typed and untyped phases of evaluation.

Catarina Coquand’s work [14] is the most closely related to our development. She formally proves soundness and completeness of NbE for simple type theory, however she considers a nameful syntax with explicit substitutions while we use intrinsic de Bruijn variables with implicit substitutions.

Altenkirch and Chapman [2] formalize a big-step normalization algorithm in Agda. This development uses an explicit substitution calculus and implements normalization using an environment machine with first-order closures. The algorithm works similarly to ours on the common syntactic fragment, but it is not structurally recursive and hence specifies evaluation as a inductive relation, using the Bove-Capretta method [10].

Altenkirch and Kaposi [4] use a glued presheaf model (“presheaf logical predicate”) for NbE for a minimal dependent type theory. This development provided the initial inspiration for the formalization in this thesis; it seemed plausible that “scaling down” dependently typed NbE to simple type theory would yield a formalization that is more compact than existing ones. This turned out to be the case; however the discussion of resulting development is relegated to Chapter 5.1, because using a Kripke model has the advantage of clean separation of the actual algorithm and its naturality proofs, which was deemed preferable to the somewhat less transparent presheaf construction. Also, our work uses a simple presheaf model rather than the glued model used in *Ibid.* or in the work of Altenkirch, Hoffman and Streicher [3].

Chapter 2

Metatheory

In this chapter the metatheory used for formalization is presented, first broadly, then its specific implementation in Agda.

2.1 Type Theory

The basic system we use is intensional Martin-Löf type theory (MLTT). For an overview and tutorial see the first chapter of the Homotopy Type Theory book [28]. However, there is no canonical definition of intensional type theory. There are numerous variations concerning type universes and the range of allowed type definitions. Our development uses the following features:

- *Strictly positive inductive definitions.* These are required for an intrinsically well-typed syntax. We do not require induction-recursion, induction-induction, higher induction or large inductive types.
- *Two predicative universes, named \mathbf{Set}_0 and \mathbf{Set}_1 , with large elimination into \mathbf{Set}_0 .* \mathbf{Set}_0 is the base universe in Agda, i. e. the universe of small types. Large elimination is needed for recursive definitions of semantic types, since inductive

definitions for them would not be positive and hence would be illegal. We only need `Set1` to type the large eliminations. In Agda `Set` is a synonym for `Set0`.

- *Function extensionality as an axiom.* This posits that two functions are equal if they take equal arguments to equal results. We could eschew it by using setoid reasoning with semantic equivalence, as C. Coquand does [14]. However, we only use function extensionality in correctness proofs, and the normalization function does not refer to it or to any other postulate. We are content with *logical* as opposed to computational content for correctness proofs. This way, the numerous congruence lemmas needed for setoid reasoning can be skipped. Also, function extensionality has computational interpretation in cubical [13] and observational [6] type theories, to which this development could be plausibly ported in the future (when practical implementations of the mentioned theories become available).

In Agda 2.5.2, there is robust support for dependent pattern matching without Streicher’s K axiom [26]. We use the `--without-K` language option for all of our code in the main development. However, we do use axiom K in Chapter 5.1 for the direct presheaf model, but only for technical convenience, as it can be avoided by additional uniqueness proofs for equalities.

2.2 Agda

Agda is a dependently typed programming language and proof assistant. Its design and core features are described in Ulf Norell’s thesis [23]. The latest documentation is available online [17]. For a book-length introduction geared towards beginners, see [27]. We nonetheless summarize here the core constructions.

2.2.1 Basic Constructions

Two essential artifacts in Agda code are *inductive type definitions* and *function definitions*.

Inductive definitions introduce new types. Agda has *open type universes*, which means that programmers may freely add new types as long as they preserve logical consistency¹. An inductive definition involves a type constructor declaration followed by zero or more term constructor declarations. For example, the type of natural numbers is defined the following way:

```
data ℕ : Set where
  zero : ℕ
  suc   : ℕ → ℕ
```

Inductive definitions may have *parameters* and *indices*. The former are implicitly quantified over the term constructors, but must be uniform in the constructors' return types. The latter must be explicitly quantified in term constructors, but are allowed to vary. The definition for length-indexed vectors exhibits both:

```
data Vec (A : Set) : ℕ → Set where
  nil   : Vec A zero
  cons  : {n : ℕ} → A → Vec A n → Vec A (suc n)
```

In a type constructor declaration, parameters are listed left to the colon, while indices are to the right. `A : Set` is a parameter, so it is implicitly quantified and is the same in the return types of `nil` and `cons`. In contrast, the length index is quantified in `cons`. We can use brackets to make parameters implicit; here `cons` has an implicit first parameter, and can be used like `cons zero nil` for a value of `Vec ℕ (suc zero)`. Implicit arguments are filled in by Agda's unification algorithm. Alternatively, the `∀` symbol can be used to leave the types of parameters implicit, and we could define `cons` as follows:

```
cons : ∀ {n} → A → Vec A n → Vec A (suc n)
```

Function definitions are given by pattern matching. In Agda, pattern matching

¹Agda checks *strict positivity* as a sufficient condition for consistency.

implements branching evaluation as it is usual in functional languages, but it also refines type indices based on particular selected cases. The main practical difference between parameters and indices is that we can gain information about indices by pattern matching, while we can't infer anything about parameters just by having a parameterized term.

For example, in the following definition of concatenation for **Vec**, the result type is refined depending on whether the first argument is empty:

```

_+_ : ℕ → ℕ → ℕ
zero + m = m
suc n + m = suc (n + m)

_++_ : ∀ {A n m} → Vec A n → Vec A m → Vec A (n + m)
nil    ++ ys = ys
cons x xs ++ ys = cons x (xs ++ ys)

```

In the `cons x xs ++ ys` case, the result type is refined to `Vec A (suc (n + m))`, allowing us to build a `suc`-long result on the right hand side with `cons`.

Agda function definitions must be total, hence pattern matching must be exhaustive and recursive calls must be structurally decreasing. Dependent pattern matching with structural recursion allows us to write the same proofs as inductive eliminators, but with a more convenient interface and less administrative burden. Also, injectivity and disjointness of term constructors is implicit in pattern matching, while they have to be separately proven when using eliminators.

As a notational convention, we shall take cues from the Idris [11] programming language and sometimes leave implicit parameters implicit even in type declarations of functions or constructors. In this style, declaring `_++_` looks as follows:

```

_++_ : Vec A n → Vec A m → Vec A (n + m)

```

This is not valid Agda, but we shall do this whenever the types and binding status of parameters are obvious.

2.2.2 Standard Library

Our development does not have any external dependencies. We need less than a hundred lines of code from the Agda standard library [18], but we choose to include it along the project in `Lib.agda`, for portability, and also make some changes.

Most of `Lib.agda` concerns equality and equational reasoning. Propositional equality is the same as in standard Agda:

```
data _≡_ {i}{A : Set i} (x : A) : A → Set i where
  refl : x ≡ x
```

We also postulate function extensionality, for functions with both implicit and explicit arguments (the “i” in `fexti` stands for “implicit”):

```
postulate
  fext  : (∀ x → f x ≡ g x) → f ≡ g
  fexti : (∀ x → f {x} ≡ g {x}) → (λ {x} → f {x}) ≡ (λ {x} → g {x})
```

Note that the types of `f` and `g` are left implicit above, in accordance with our notational liberties. In full Agda we write:

```
fext :
  ∀ {i j}{A : Set i}{B : A → Set j}{f g : (x : A) → B x}
  → ((x : A) → f x ≡ g x) → f ≡ g
```

Here the quantification noise is increased by universe indices. Universe polymorphism is only used in `Lib.agda`, but we will henceforth omit universe indices anyway.

Our style of equational reasoning deviates from the standard library. First, we use a more compact notation for symmetry and transitivity, inspired by the Homotopy Type Theory book [28]:

```
_■_ : x ≡ y → y ≡ z → x ≡ z -- transitivity
refl ■ refl = refl

_⁻¹ : x ≡ y → y ≡ x -- symmetry
refl ⁻¹ = refl
```

$_^{-1}$ is postfix, so if we have $(p : x \equiv y)$, then (p^{-1}) has type $(y \equiv x)$.

Second, for reasoning about congruences we mainly use two operations. The first (named `ap` in the HoTT book and `cong` in the standard library) expresses that functions respect equality:

```
_&_ : (f : A → B) → x ≡ y → f x ≡ f y
f & refl = refl
infixl 9 _&_
```

The second one expresses that *non-dependent function application* respects equality as well:

```
_⊗_ : f ≡ g → x ≡ y → f x ≡ g y
refl ⊗ refl = refl
infixl 8 _⊗_
```

Using these two operators, we can lift non-dependent functions with arbitrary arities to congruences. For example, if we have $(p : a_0 \equiv a_1)$, $(q : b_0 \equiv b_1)$ and $(f : A \rightarrow B \rightarrow C)$, then $(f \& p \otimes q)$ has type $(f a_0 b_0 \equiv f a_1 b_1)$. Note that both operators associate left, and `_&_` binds more strongly. This is reminiscent of the lifting syntax with applicative functors [22] in Haskell programming.

Furthermore, we use coercion instead of transport (see HoTT book [28, pp. 72], also named `subst` in the Agda standard library):

```
coe : A ≡ B → A → B
coe refl a = a
```

`transport/subst` can be recovered by `_&_` and `coe`:

```
subst : (P : A → Set) → x ≡ y → P x → P y
subst P eq = coe (P & eq)
```

The choice for `coe` is mainly stylistic and rather subjective. `coe (P & eq)` does not look worse than `subst P eq`, but `coe` by itself is used often and is more compact than `subst id`.

In Agda, sometimes an explicit “equational reasoning” syntax is used. For example,

commutativity for addition may look like the following:

```

+-comm : (m n : ℕ) → m + n ≡ n + m
+-comm zero    n = +-right-identity n -1
+-comm (suc m) n = begin
  suc m + n    ≡⟨ ⟩
  suc (m + n)  ≡⟨ suc & +-comm m n ⟩
  suc (n + m)  ≡⟨ +-suc n m -1 ⟩
  n + suc m    QED

```

In formal development we do not use this style and instead keep the intermediate expressions implicit:

```

+-comm (suc m) n = (suc & +-comm n m) ■ (+-suc n m -1)

```

This is because in an interactive environment one can step through the intermediate points fairly easily, and writing them out adds significant visual clutter. However, we will use informal equational reasoning when it aids explanation.

We also borrow Σ (dependent sum), τ (unit type), \perp (empty type) and $_ \sqcup _$ (disjoint union) from the standard library. Elements of Σ are constructed as (a, b) pairs, and the projections are `proj1` and `proj2`. τ has `tt` (“trivially true”) as sole element. Injection into disjoint unions is done with `inj1` and `inj2`. The *ex falso* principle is named `⊥-elim`, and we have `⊥-elim : {A : Set} → ⊥ → A`.

Chapter 3

Syntax

In this chapter we present the syntax of simply typed lambda calculus - in our case intrinsic syntax with implicit substitutions and de Bruijn indices. We take term conversion to be part of the syntax and therefore include it in this chapter.¹ Since the definition of conversion refers to substitution and weakening, they are also presented here.

3.1 Base Syntax

The complete definition is the following:

```
data Ty : Set where
  ι      : Ty
  _⇒_    : Ty → Ty → Ty

data Con : Set where
  •      : Con
  _,_    : Con → Ty → Con
```

¹We take the view that the “proper” definition of the syntax would be an initial object in some suitable category of models, e. g. the initial category with families. In this way, models include conversion as propositional equations, so the syntax is quotiented by conversion as well; see e. g. [12] and [5]. We do not use this definition because quotients are not yet natively supported in Agda, and we find value in developing a conventional presentation first.

```

data _∈_ (A : Ty) : Con → Set where
  vz : A ∈ (Γ , A)
  vs : A ∈ Γ → A ∈ (Γ , B)

data Tm Γ : Ty → Set where
  var : A ∈ Γ → Tm Γ A
  lam : Tm (Γ , A) B → Tm Γ (A ⇒ B)
  app : Tm Γ (A ⇒ B) → Tm Γ A → Tm Γ B

```

There is a base type ι (Greek iota) and function types. Contexts are just lists of types. $_ \in _$ is an inductively defined membership relation on contexts. It has the same structure as de Bruijn indices; for this reason they are named **vz** for “zero” and **vs** for “suc”. Terms are parameterized by a Γ context and indexed by a type. **var** picks an entry from the contexts, **lam** is abstraction while **app** is function application.

As an example, the identity function on the base type is represented as:

```

id-ι : Tm • (ι ⇒ ι)
id-ι = lam (var vz) -- with nameful syntax: λ (x : ι) . x

```

The presented syntax is *intrinsic*. In other words, only the well-typed terms are defined. An *extrinsic* definition would first define untyped *preterms* and separately a typing relation on preterms and contexts:

```

data Tm : Set where -- omitted constructors
data _⊢_ (Γ : Con) → Tm → Ty → Set where -- omitted constructors

```

Now, $\Gamma \vdash t \in A$ expresses that the preterm t is well-formed and has type A in context Γ .

There are advantages and disadvantages to both intrinsic and extrinsic definitions. With intrinsic syntax, well-formedness and type preservation come for free. On the other hand, proofs and computations which do not depend on types are easier to formulate with preterms. In the case of STLC, the type system is simple enough so that carrying around type information never becomes burdensome, so in this setting intrinsic definitions are all-around more convenient. For polymorphic systems such as System

F , intrinsic typing introduces significant bureaucratic overhead, by tagging terms with type coercions; see Benton et al. [8] for approaches to dealing with them. For extrinsic syntax, powerful automation for reasoning about substitution is available in Coq, allowing spectacularly compact proofs [25]. It remains to be seen if it can be practically adapted to intrinsic syntax. For dependent type theories, intrinsic typing with quotient inductive-inductive definitions seems to be most promising, as it requires a relatively compact set of rules [5], in contrast to extrinsic approaches which suffer from a veritable explosion of well-formedness conditions and congruence rules.

3.2 Context Embeddings and Substitutions

In order to specify β -conversion, we need a notion of substitution. Also, the η -rule must refer to a notion of *weakening* or *embedding*: it mentions the “same” term under and over a λ binder, but the two mentions cannot be definitionally the same, since they are in different contexts and thus have different types. We need to express the notion that whenever one has a term in a context, one can construct essentially the same term in a larger context. We use *order-preserving-embeddings* for this. Embeddings enable us to state the η -rule, but we also make use of them for defining substitutions.

3.2.1 Embeddings

We define order-preserving embeddings the following way:

```
data OPE : Con  $\rightarrow$  Con  $\rightarrow$  Set where
  •   : OPE • •
  drop : OPE  $\Gamma$   $\Delta \rightarrow$  OPE ( $\Gamma$  , A)  $\Delta$ 
  keep  : OPE  $\Gamma$   $\Delta \rightarrow$  OPE ( $\Gamma$  , A) ( $\Delta$  , A)
```

Elements of $\text{OPE } \Gamma \Delta$ express that Δ can be obtained from Γ by dropping zero or more entries in order.

Some choices can be made here. C. Coquand [15] uses an explicit identity embedding

constructor with type $(\forall \{\Gamma\} \rightarrow \text{OPE } \Gamma \ \Gamma)$ instead of our \bullet . This slightly simplifies some proofs and slightly complicates others. It does not have propositionally unique identity embeddings², but overall the choice between this definition and ours is fairly arbitrary.

An alternative solution is using *renamings*. Renamings are essentially substitutions containing only variables, alternatively definable as functions mapping variables of smaller contexts to variables of larger contexts [21, pp. 24]:

$$\begin{aligned} \text{Ren} &: \text{Con} \rightarrow \text{Con} \rightarrow \text{Set} \\ \text{Ren } \Gamma \ \Delta &= \{A : \text{Ty}\} \rightarrow A \in \Delta \rightarrow A \in \Gamma \end{aligned}$$

However, renamings are able to express *permutations* of contexts as well, which we have no need for in this development. Thus, renamings are larger than what is strictly necessary, which may or may not result in technical difficulties. We aim to choose the smallest workable representation, as a general principle.

Embeddings have action on variables and terms, reconstructing them in larger contexts. We denote embedding by lowercase “e” subscripts:

$$\begin{aligned} \epsilon_e &: \text{OPE } \Gamma \ \Delta \rightarrow A \in \Delta \rightarrow A \in \Gamma \\ \epsilon_e \bullet &= v \\ \epsilon_e (\text{drop } \sigma) v &= v \sigma (\epsilon_e \sigma v) \\ \epsilon_e (\text{keep } \sigma) vz &= vz \\ \epsilon_e (\text{keep } \sigma) (vs v) &= vs (\epsilon_e \sigma v) \\ \\ \text{Tm}_e &: \text{OPE } \Gamma \ \Delta \rightarrow \text{Tm } \Delta \ A \rightarrow \text{Tm } \Gamma \ A \\ \text{Tm}_e \sigma (\text{var } v) &= \text{var } (\epsilon_e \sigma v) \\ \text{Tm}_e \sigma (\text{lam } t) &= \text{lam } (\text{Tm}_e (\text{keep } \sigma) t) \\ \text{Tm}_e \sigma (\text{app } f \ a) &= \text{app } (\text{Tm}_e \sigma f) (\text{Tm}_e \sigma a) \end{aligned}$$

Identity embeddings keep every entry:

$$\begin{aligned} \text{id}_e &: \forall \{\Gamma\} \rightarrow \text{OPE } \Gamma \ \Gamma \\ \text{id}_e \{\bullet\} &= \bullet \\ \text{id}_e \{\Gamma, A\} &= \text{keep } (\text{id}_e \{\Gamma\}) \end{aligned}$$

We define *wk* as shorthand for the embedding that only drops the topmost entry:

$$\text{wk} : \forall \{A \ \Gamma\} \rightarrow \text{OPE } (\Gamma, A) \ \Gamma$$

²Since wrapping the identity embedding with any number of *keep*-s also yields identity embeddings.

```
wk = drop ide
```

3.2.2 Substitutions

A salient feature of our syntax is the lack of explicit substitution, i. e. we define substitutions separately as lists of terms, and their action on terms is given as a recursive function. We choose implicit substitutions for two reasons:

1. It is the canonical presentation of STLC in textbooks [24, 19].
2. Lighter formalization. With implicit substitutions, the definition of conversion becomes significantly smaller. With explicit substitutions, we need congruence closure of β and η conversion on terms and substitutions, plus rules for the action of substitution on terms. With implicit substitutions, only terms need to be considered. For illustration, Altenkirch and Chapman’s definition [2] of conversion for explicit substitution calculus has twenty-four rules, while this development has only seven. It also follows that with our syntax, equational reasoning about substitutions involves definitional or propositional equalities, which are automatically respected by all constructions. There is minor complication involving semantic interpretation of substitutions, discussed in Chapter 4.7.

Strictly speaking, stating β -conversion only requires single substitutions. However, simultaneous substitution is easier to define and reason about. Thus, we shall define the former using the latter and identity substitutions. We have substitutions as lists of terms:

```
data Sub ( $\Gamma$  : Con) : Con  $\rightarrow$  Set where
  •   : Sub  $\Gamma$  •
  _,_ : Sub  $\Gamma$   $\Delta \rightarrow$  Tm  $\Gamma$  A  $\rightarrow$  Sub  $\Gamma$  ( $\Delta$  , A)
```

Elements of $\text{Sub } \Gamma \Delta$ contain a $\text{Tm } \Gamma A$ for each A type in Δ . In other words, an element of $\text{Sub } \Gamma \Delta$ assigns to each variable in Δ a term in Γ .

Again, there are some choices in defining **Sub**. Identity substitutions (which have the

identity action on terms) could be represented as an explicit constructor, as well as **Sub** composition and weakenings. Our **Sub** can be seen as a normal form of substitutions: explicit composition and weakening forms tree-like structures, but we flatten them to just lists of terms. This eliminates redundancy, and allows us to define compositions and identities recursively, which provides us with more definitional equalities.

We could also give a functional definition (see [21, pp. 24]), similarly to how we did for renamings before:

```
Sub : Con → Con → Set
Sub Γ Δ = {A : Ty} → A ∈ Δ → Tm Γ A
```

However, this contains many definitionally distinct functions with the same action on terms, and so we dismiss it on grounds of unnecessary largeness.

Before we can define identity substitutions and the action of substitution on terms, we need an operation with type $(\forall \{A \in \Delta\} \rightarrow \text{Sub } \Gamma \Delta \rightarrow \text{Sub } (\Gamma, A) (\Delta, A))$. This enables pushing substitutions under λ -binders when recursing on terms. Constructing this substitution requires that we embed all terms in the input substitution into the extended (Γ, A) context. We define this as right composition with an embedding:

```
_s ∘ e_ : Sub Δ Σ → OPE Γ Δ → Sub Γ Σ
•      s ∘ e δ = •
(σ , t) s ∘ e δ = (σ s ∘ e δ) , Tm_e δ t
```

The notation $_s \circ e_$ expresses that we have a substitution on the left and an embedding on the right. Its type resembles that of ordinary function composition, if we consider both **Sub** and **OPE** as context morphisms. We will expand on the categorical interpretation of this in Chapter 4.4.

There is a canonical injection $\ulcorner_ \urcorner^{\text{OPE}}$ from **OPE** to **Sub**:

```
drop_s : Sub Γ Δ → Sub (Γ , A) Δ
drop_s σ = σ s ∘ e wk

keep_s : Sub Γ Δ → Sub (Γ , A) (Δ , A)
keep_s σ = drop_s σ , var vz
```

$$\begin{aligned}
&\vdash \text{ }_{\text{ope}} : \text{OPE } \Gamma \Delta \rightarrow \text{Sub } \Gamma \Delta \\
&\vdash \bullet \quad \text{ }_{\text{ope}} = \bullet \\
&\vdash \text{drop } \sigma \text{ }_{\text{ope}} = \text{drop}_s \vdash \sigma \text{ }_{\text{ope}} \\
&\vdash \text{keep } \sigma \text{ }_{\text{ope}} = \text{keep}_s \vdash \sigma \text{ }_{\text{ope}}
\end{aligned}$$

We note that keep_s is the needed operation for pushing substitutions under binders.

Now the action of substitution on terms and variables can be defined:

$$\begin{aligned}
&\epsilon_s : \text{Sub } \Gamma \Delta \rightarrow A \in \Delta \rightarrow \text{Tm } \Gamma A \\
&\epsilon_s (\sigma, t) \text{ vz} = t \\
&\epsilon_s (\sigma, t) (\text{vs } v) = \epsilon_s \sigma v \\
&\text{Tm}_s : \text{Sub } \Gamma \Delta \rightarrow \text{Tm } \Delta A \rightarrow \text{Tm } \Gamma A \\
&\text{Tm}_s \sigma (\text{var } v) = \epsilon_s \sigma v \\
&\text{Tm}_s \sigma (\text{lam } t) = \text{lam } (\text{Tm}_s (\text{keep}_s \sigma) t) \\
&\text{Tm}_s \sigma (\text{app } f a) = \text{app } (\text{Tm}_s \sigma f) (\text{Tm}_s \sigma a)
\end{aligned}$$

ϵ_s simply looks up a term from a substitution³, while Tm_s recurses into terms to substitute all variables. Identity substitutions can be defined as well:

$$\begin{aligned}
&\text{id}_s : \{\Gamma : \text{Con}\} \rightarrow \text{Sub } \Gamma \Gamma \\
&\text{id}_s \{\bullet\} = \bullet \\
&\text{id}_s \{\Gamma, A\} = \text{keep}_s \text{id}_s
\end{aligned}$$

Single substitution with a $(t : \text{Tm } \Gamma A)$ term is given by (id_s, t) . This assigns the t term to the zeroth de Bruijn variable and leaves all other variables unchanged.

3.3 Conversion

The conversion relation is given as:

³The immediate reason for not naming ϵ_s simply “lookup” is that we will need several different lookup functions for various purposes. For any sizable formal development, naming schemes eventually emerge, for better or worse. The author has found that uniformly encoding salient information about types of operations in their names is worthwhile to do. It is also not easy to hit the right level of abstraction; too little and we repeat ourselves too much or neglect to include known structures, but too much abstraction may cause the project to be less accessible and often also less convenient to develop in the first place. Many of the operations and proofs here could be defined explicitly using categorical language, i. e. bundling together functors’ actions on objects, morphisms and category laws in records, then projecting out required components. Our choice is to keep the general level of abstraction low for this development.

```

data _~_ {Γ} : ∀ {A} → Tm Γ A → Tm Γ A → Set where
  η      : t ~ lam (app (Tme wk t) (var vz))
  β      : app (lam t) t' ~ Tms (ids , t') t

  lam    : t ~ t' → lam t ~ lam t'
  app    : f ~ f' → a ~ a' → app f a ~ app f' a'

  ~refl  : t ~ t
  _~-1_ : t ~ t' → t' ~ t
  _~■_   : t ~ t' → t' ~ t'' → t ~ t''

```

η and β are the actual conversion rules. We push t under a lambda with $(Tm_e \text{ wk})$ in η , and use single substitution for β . The rest are rules for congruence (lam , app) and equivalence closure ($\sim\text{refl}$, \sim^{-1} , \sim^\blacksquare). The syntax for equivalence rules is the same as for propositional equality, except for the \sim prefixes.

Chapter 4

Normalization

In this chapter we specify normal forms, then implement normalization and prove its correctness.

4.1 Normal terms

Naturally, we need to know what normal forms are, if we are to normalize. Our definition is entirely standard: normal terms are either lambdas or *neutral* terms of base type, and neutral terms are variables applied to zero or more normal arguments:

```
mutual
data Nf (Γ : Con) : Ty → Set where
  ne  : Ne Γ ⊔ → Nf Γ ⊔
  lam : Nf (Γ , A) B → Nf Γ (A ⇒ B)

data Ne (Γ : Con) : Ty → Set where
  var : A ∈ Γ → Ne Γ A
  app : Ne Γ (A ⇒ B) → Nf Γ A → Ne Γ B
```

Agda allows us to overload `var`, `lam` and `app` for normal forms. β -normality is obvious, since only variables can be applied. Normal forms are also η -long: since `ne` only injects neutrals of base type, all normal terms of function type must in fact be lambdas. Only

requiring β -normality would be as simple as having $(ne : \forall \{A\} \rightarrow Ne \ \Gamma \ A \rightarrow Nf \ \Gamma \ A)$. Alternatively, normal forms can be given as

```
data Nf (Γ : Con) : (A : Ty) → Tm Γ A → Set
```

with the same construction as before except that it is a predicate on general terms, expressing their normality. This definition would be roughly as convenient as the one we use. We also need actions of context embeddings:

```
Nfe : OPE Γ Δ → Nf Δ A → Nf Γ A -- definitions omitted
Nee : OPE Γ Δ → Ne Δ A → Ne Γ A -- definitions omitted
```

These are defined by straightforward mutual recursion.

4.2 Preliminaries: Standard and Enriched Models

Clearly, STLC is a small fragment of Agda, so we should be able to interpret the syntax back to Agda types and constructions in a straightforward way. From a semantic viewpoint, the most straightforward interpretation of the syntax is called the *standard model*. From an operational viewpoint, the standard model is just a well-typed interpreter [7] for STLC as an embedded language. It is implemented as follows:

```
Tys : Ty → Set
Tys ⊥      = ⊥
Tys (A ⇒ B) = Tys A → Tys B

Cons : Con → Set
Cons •      = ⊤
Cons (Γ , A) = Cons Γ × Tys A

∈s : ∀ {Γ A} → A ∈ Γ → (Cons Γ → Tys A)
∈s vz      Γs = proj2 Γs
∈s (vs v) Γs = ∈s v (proj1 Γs)

Tms : ∀ {Γ A} → Tm Γ A → (Cons Γ → Tys A)
Tms (var v)   Γs = ∈s v Γs
Tms (lam t)   Γs = λ as → Tms t (Γs , as)
Tms (app f a) Γs = Tms f Γs (Tms a Γs)
```

Traditionally, notation for semantics uses double brackets, e. g. the interpretation of types would look like this:

```

[[_]] : Ty → Set
[[ ⊥   ]] = ⊥
[[ A ⇒ B ]] = [[ A ]] → [[ B ]]

```

We instead opt for notating semantic interpretations with superscripts on type constructors (contrast *subscripts*, which we use for denoting operations for embeddings and substitutions). Syntactic overloading in Agda is not convenient for our purposes here, so we have to distinguish different constructions (terms or types, etc.) and different models in any case. Therefore it makes sense to just reuse the names of types in the syntax, and distinguish different models by different superscripts. Similarly, we annotate names of function arguments with superscripts, if they are elements of certain semantic sets. For example, for some model \mathbb{M} we use $\Gamma^{\mathbb{M}}$, $\Delta^{\mathbb{M}}$ etc. for elements of semantic contexts, and $\mathbf{a}^{\mathbb{M}}$, $\mathbf{b}^{\mathbb{M}}$ etc. for semantic terms.

As to the implementation of the model: we interpret the base type with the empty type, and functions as functions. Contexts are interpreted as lists of semantic values. Terms are interpreted as functions from semantic contexts to semantic types. Now, normalizing $(\text{tm}^s \ (\text{lam} \ (\text{var} \ \text{vz})) \ \text{tt})$ yields the $(\lambda \ \mathbf{a}^s \rightarrow \mathbf{a}^s)$ Agda term, so indeed we interpret a syntactic identity function with a metatheoretic identity function.

The standard model already has a key feature of NbE: it uses metatheoretical functions for evaluation. However, the standard model does not allow one to go back to syntax from semantics. If we have an $(\mathbf{f} : \mathbf{A}^s \rightarrow \mathbf{B}^s)$ semantic function, all we can do with it is to apply it to an argument and extract a \mathbf{B}^s . *Weak evaluation* of programs assumes that all variables are mapped to semantic values. This corresponds to the usual notion of program evaluation in common programming languages. In contrast, strong normalization requires reducing terms inside function bodies, under binders. Bound variables are not mapped to any semantic value, so they block evaluation. In STLC, variables block function application; in richer systems variables may block case analysis or elimination of inductively defined data as well. See [1, pp. 12] for an overview for

various approaches for implementing computation under binders.

NbE adds additional structure to semantic types, internalizing computation with blocking variables, which allows one to get back to syntax, moreover, to a subset of syntax containing only normal terms.

Adding progressively more structure to models allows us to prove more properties about the syntax. The standard model yields a simple proof of consistency, namely that there is no term with base type in the empty context:

```
consistency : Tm • 1 → 1
consistency t = Tms t tt
```

Kripke models contain slightly more structure than the standard model, allowing us to implement normalization, which is a *completeness* theorem from a logical viewpoint [16], as per the Curry-Howard correspondence. Adding yet more structure yields *presheaf models*, which enable correctness proofs for normalization as well. We summarize the computational and logical interpretations below.

Table 4.1: Overview of models

Model	Computation	Logical proof
Standard	Type-safe interpreter	Consistency
Kripke	Normalizer	Completeness
Presheaf	Normalizer	Proof-relevant completeness

4.3 Normalization

First, we present the implementation in its entirety. Then, we briefly discuss Kripke models and how ours is a particular instance of them.

We denote the model for normalization with capital “N” superscript. The implementation follows a similar shape as the standard model. The key difference - from which

others follow - is in the interpretation of functions:

$$\begin{aligned} \text{Ty}^N &: \text{Ty} \rightarrow \text{Con} \rightarrow \text{Set} \\ \text{Ty}^N \text{ } \iota & \quad \Gamma = \text{Nf } \Gamma \text{ } \iota \\ \text{Ty}^N (A \Rightarrow B) \Gamma &= \forall \{\Delta\} \rightarrow \text{OPE } \Delta \Gamma \rightarrow \text{Ty}^N A \Delta \rightarrow \text{Ty}^N B \Delta \end{aligned}$$

Each type is mapped to a $\text{Con} \rightarrow \text{Set}$ predicate. Hence, semantic types are not just sets anymore, but families of sets indexed by contexts. The base type is mapped to the family of its normal forms. Functions are mapped to semantic functions which take semantic inputs to semantic outputs in *any context larger than Γ* . The additional OPE parameter is the key to the algorithm. As it was noted in Chapter 4.2, semantic functions must be able to take blocking variables as inputs. Blocking variables must be necessarily “fresh” with respect to the Γ context of a semantic function. Here, we can apply a semantic function to suitable embedding into a larger context, allowing to subsequently apply it to semantic terms containing variables pointing into that context. In the simplest case (and the only case we will actually use), if we have

$$f^N : \forall \{\Delta\} \rightarrow \text{OPE } \Delta \Gamma \rightarrow \text{Ty}^N A \Delta \rightarrow \text{Ty}^N B \Delta$$

then we also have

$$f^N (\text{wk } \{A\}) : \text{Ty}^N A (\Gamma, A) \rightarrow \text{Ty}^N B (\Gamma, A)$$

The interpretations of contexts and variables are analogous to those in the standard model: contexts become lists of semantic values and variables look up semantic values from semantic contexts. However, we define semantic contexts inductively rather than recursively, for technical convenience.

$$\begin{aligned} \text{data } \text{Con}^N &: \text{Con} \rightarrow \text{Con} \rightarrow \text{Set} \text{ where} \\ &\bullet : \text{Con}^N \bullet \Delta \\ _ , _ &: \text{Con}^N \Gamma \Delta \rightarrow \text{Ty}^N A \Delta \rightarrow \text{Con}^N (\Gamma, A) \Delta \\ \epsilon^N &: \forall \{\Gamma A\} \rightarrow A \in \Gamma \rightarrow \forall \{\Delta\} \rightarrow \text{Con}^N \Gamma \Delta \rightarrow \text{Ty}^N A \Delta \\ \epsilon^N \text{ vz} & \quad (\Gamma^N, t^N) = t^N \\ \epsilon^N (\text{vs } v) & \quad (\Gamma^N, _) = \epsilon^N v \Gamma^N \end{aligned}$$

When interpreting terms, variables and applications are easy. In the latter case we just apply the recursive result to id_e - we can choose *not* to conjure up fresh variables

if there is no need to. We run into a bit of a bump in the interpretation of lambdas:

```

TmN : ∀ {Γ A} → Tm Γ A → ∀ {Δ} → ConN Γ Δ → TyN A Δ
TmN (var v)   ΓN = ∈N v ΓN
TmN (lam t)   ΓN = λ σ aN → TmN t (? , aN)  -- ? marks the bump
TmN (app f a) ΓN = TmN f ΓN ide (TmN a ΓN)

```

Clearly, the idea is to evaluate the inner t term in (Γ^N, a^N) as we did in the standard model, but Γ^N does not have the right type. It has type $\text{Con}^N \Gamma \Delta$, but we need to return in some Σ extended context, with $(\sigma : \text{OPE } \Sigma \Delta)$ witnessing the extension. So, what we need is an action of context embedding on semantic contexts, and on semantic terms as well, since the former are just lists of the latter. To implement it, we first need *composition* for context embeddings:

```

_◦e_ : OPE Δ Σ → OPE Γ Δ → OPE Γ Σ
σ      ◦e •          = σ
σ      ◦e drop δ = drop (σ ◦e δ)
drop σ ◦e keep δ = drop (σ ◦e δ)
keep σ ◦e keep δ = keep (σ ◦e δ)

```

It witnesses transitivity for the embedding relation. Then, embeddings on semantic terms is easy enough:

```

TyNe : OPE Δ Γ → TyN A Γ → TyN A Δ
TyNe {ι}      σ tN = Nfe σ tN
TyNe {A ⇒ B} σ tN = λ δ aN → tN (σ ◦e δ) aN

ConNe : OPE Σ Δ → ConN Γ Δ → ConN Γ Σ
ConNe σ •          = •
ConNe σ (ΓN , tN) = ConNe σ ΓN , TyNe σ tN

```

For $\text{Ty}^{N_e} \{\iota\}$, we use embedding of normal forms. For $\text{Ty}^{N_e} \{A \Rightarrow B\}$, we compose the externally given σ embedding with the input embedding δ . Con^{N_e} is just pointwise application of Ty^{N_e} . Tm^N can be now given as:

```

TmN : ∀ {Γ A} → Tm Γ A → ∀ {Δ} → ConN Γ Δ → TyN A Δ
TmN (var v)   ΓN = ∈N v ΓN
TmN (lam t)   ΓN = λ σ aN → TmN t (ConNe σ ΓN , aN)
TmN (app f a) ΓN = TmN f ΓN ide (TmN a ΓN)

```

We still need a `quote` function to transform semantic terms to normal terms. We

also need to mutually define an `unquote` function which sends neutral terms to semantic terms, for reasons shortly illuminated. Then, normalization is given as evaluation followed by quotation, where evaluation uses a semantic context given by unquoting each variable in the context.

mutual

```

qN : ∀ {A Γ} → TyN A Γ → Nf Γ A
qN {ι}      tN = tN
qN {A ⇒ B} tN = lam (qN (tN wk (uN (var vz))))

uN : ∀ {A Γ} → Ne Γ A → TyN A Γ
uN {ι}      n = ne n
uN {A ⇒ B} n = λ σ aN → uN (app (Nee σ n) (qN aN))

uc N : ∀ {Γ} → ConN Γ Γ
uc N {•}      = •
uc N {Γ , A} = ConNe wk uc N , uN (var vz)

nf : ∀ {Γ A} → Tm Γ A → Nf Γ A
nf t = qN (TmN t uc N)

```

At first, it can be difficult to build a mental model of the operation of the algorithm. On the highest level, normalization alternates between evaluation and quoting: a term is first evaluated, then if it is a function, its semantic function is applied to a “blocking” input and we quote the result, which can be a semantic function again. This evaluate-quote alternation proceeds until the result is not a function but a base value. Note that although q^N does not directly call Tm^N , it does apply semantic functions, which may *internally* call Tm^N .

In fact, this encapsulation of recursive Tm^N calls is a crucial detail which makes this whole definition structurally recursive and thus total. In the $(Tm^N (\text{lam } t) \Gamma^N = \lambda \sigma a^N \rightarrow Tm^N t (Con^N_e \sigma \Gamma^N , a^N))$ clause, the recursive Tm^N call is evidently structural, and because it happens under a *metatheoretic* lambda, semantic functions can be applied to *any* value in any definition without ruining structurality.

To explain the previous scare quotes around “blocking”: in the $(t^N \text{ wk } (u^N (\text{var } vz))))$

application, the $(u^N \text{ (var } vz))$ term plays the role of blocking input, but it is not quite entirely blocking. u^N performs an operation best described as *semantic η -expansion*: it acts as identity on neutral base terms, and from neutral function it produces semantic function which build up neutral applications from their inputs. Thus, $(u^N \{i\} \text{ (var } vz))$ simply reduces to $(ne \text{ (var } vz))$, while $(u^N \{i \Rightarrow i\} \text{ (var } vz))$ reduces to $(\lambda \sigma a^N \rightarrow ne \text{ (app (var } (\epsilon_e \sigma vz)) a^N))$. In short, u^N returns semantic values which will yield properly η -expanded normal forms when quoted. As convoluted this may seem, it is actually a very elegant solution.

In the section introduction we have stated that the N superscript specifies a Kripke model of STLC. What are Kripke models, then?

4.4 Laws for Embedding and Substitution

TODO

4.5 Completeness

TODO

4.6 Presheaf Model

TODO

4.7 Soundness

TODO

4.8 Stability

TODO

Chapter 5

Variations

5.1 Direct Presheaf Model

- Definition, proofs
- Semantic equality == propositional equality
- However: funext in definition of nf. Use external canonicity proof or metatheory with funext

5.2 More Efficient Evaluation

- efficient embedding for semantic contexts
- Agda benchmarking
- performance limitations of intrinsic syntax compared to type assignment

Chapter 6

Discussion and Future Work

- Technical overhead: Vs big-step, hereditary subst, Coquand, Abel (?)
- Linecounts vs big-step and hsubst (email C. Coquand to get sources?)
- Usability as EDSL normalizer
- Scaling up the technique
- With implicit substitution and conversion relation: to System F, probably

Bibliography

- [1] Andreas Abel. “Normalization by Evaluation: Dependent Types and Impredicativity”. Habilitation thesis. PhD thesis. Ludwig-Maximilians-Universität München, 2013.
- [2] Thorsten Altenkirch and James Chapman. “Big-step normalisation”. In: *Journal of Functional Programming* 19.3-4 (2009), pp. 311–333.
- [3] Thorsten Altenkirch, Martin Hofmann, and Thomas Streicher. “Categorical reconstruction of a reduction free normalization proof”. In: *Category Theory and Computer Science*. Springer. 1995, pp. 182–199.
- [4] Thorsten Altenkirch and Ambrus Kaposi. “Normalisation by evaluation for dependent types”. In: *LIPICs-Leibniz International Proceedings in Informatics*. Vol. 52. Schloss Dagstuhl-Leibniz-Zentrum fuer Informatik. 2016.
- [5] Thorsten Altenkirch and Ambrus Kaposi. “Type theory in type theory using quotient inductive types”. In: *ACM SIGPLAN Notices*. Vol. 51. 1. ACM. 2016, pp. 18–29.
- [6] Thorsten Altenkirch, Conor McBride, and Wouter Swierstra. “Observational equality, now!” In: *Proceedings of the 2007 workshop on Programming languages meets program verification*. ACM. 2007, pp. 57–68.
- [7] Lennart Augustsson and Magnus Carlsson. *An exercise in dependent types: A well-typed interpreter*. 1999.

- [8] Nick Benton et al. “Strongly typed term representations in Coq”. In: *Journal of automated reasoning* 49.2 (2012), pp. 141–159.
- [9] Ulrich Berger and Helmut Schwichtenberg. “An inverse of the evaluation functional for typed lambda-calculus”. In: *Logic in Computer Science, 1991. LICS’91., Proceedings of Sixth Annual IEEE Symposium on.* IEEE. 1991, pp. 203–211.
- [10] Ana Bove and Venanzio Capretta. “Modelling general recursion in type theory”. In: *Mathematical Structures in Computer Science* 15.04 (2005), pp. 671–708.
- [11] Edwin Brady. “Idris, a general-purpose dependently typed programming language: Design and implementation”. In: *Journal of Functional Programming* 23.05 (2013), pp. 552–593.
- [12] Simon Castellan. *Dependent type theory as the initial category with families*. 2014. URL: <http://iso.mor.phis.me/archives/2011-2012/stage-2012-goteburg/report.pdf> (visited on 03/10/2017).
- [13] Cyril Cohen et al. “Cubical type theory: a constructive interpretation of the univalence axiom”. In: *arXiv preprint arXiv:1611.02108* (2016).
- [14] Catarina Coquand. “A formalised proof of the soundness and completeness of a simply typed lambda-calculus with explicit substitutions”. In: *Higher-Order and Symbolic Computation* 15.1 (2002), pp. 57–90.
- [15] Catarina Coquand. “A proof of normalization for simply typed lambda calculus written in ALF”. In: *Proceedings of the Workshop on Types for Proofs and Programs*. Citeseer. 1992, pp. 85–92.
- [16] Thierry Coquand and Peter Dybjer. “Intuitionistic model constructions and normalization proofs”. In: *Mathematical Structures in Computer Science* 7.1 (1997), pp. 75–94.
- [17] Agda developers. *Agda documentation*. 2017. URL: <https://agda.readthedocs.io/en/v2.5.2/> (visited on 03/10/2017).
- [18] Agda developers. *Agda standard library*. 2017. URL: <https://github.com/agda/agda-stdlib> (visited on 03/10/2017).

- [19] Robert Harper. *Practical foundations for programming languages*. Cambridge University Press, 2016.
- [20] Per Martin-Löf. “An intuitionistic theory of types: Predicative part”. In: *Studies in Logic and the Foundations of Mathematics* 80 (1975), pp. 73–118.
- [21] Conor McBride. “Datatypes of Datatypes”. In: (2015).
- [22] Conor McBride and Ross Paterson. “Applicative programming with effects”. In: *Journal of functional programming* 18.01 (2008), pp. 1–13.
- [23] Ulf Norell. *Towards a practical programming language based on dependent type theory*. Vol. 32. Citeseer, 2007.
- [24] Benjamin C Pierce. *Types and programming languages*. MIT press, 2002.
- [25] Steven Schäfer, Tobias Tebbi, and Gert Smolka. “Autosubst: Reasoning with de Bruijn Terms and Parallel Substitutions”. In: *Interactive Theorem Proving - 6th International Conference, ITP 2015, Nanjing, China, August 24-27, 2015*. Ed. by Xingyuan Zhang and Christian Urban. LNAI. Springer-Verlag, Aug. 2015.
- [26] Thomas Streicher. *Investigations into intensional type theory*. <http://www.mathematik.tu-darmstadt.de/~streicher/HabilStreicher.pdf>, 1993.
- [27] Aaron Stump. *Verified Functional Programming in Agda*. New York, NY, USA: Association for Computing Machinery and Morgan & Claypool, 2016. ISBN: 978-1-97000-127-3.
- [28] The Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathematics*. Institute for Advanced Study: <https://homotopytypetheory.org/book>, 2013.